
ΠΑΝΕΛΛΑΔΙΚΕΣ ΕΞΕΤΑΣΕΙΣ   

ΗΜΕΡΗΣΙΟΥ ΚΑΙ ΕΣΠΕΡΙΝΟΥ ΓΕΝΙΚΟΥ ΛΥΚΕΙΟΥ 

ΤΕΤΑΡΤΗ 4 ΙΟΥΝΙΟΥ 2025 

ΕΞΕΤΑΖΟΜΕΝΟ ΜΑΘΗΜΑ: ΠΛΗΡΟΦΟΡΙΚΗ ΠΡΟΣΑΝΑΤΟΛΙΣΜΟΥ  

ΣΥΝΟΛΟ ΣΕΛΙΔΩΝ: ΕΞΙ (6) 

  

ΕΝΔΕΙΚΤΙΚΕΣ ΠΛΗΡΕΙΣ ΑΠΑΝΤΗΣΕΙΣ 

 

ΘΕΜΑ Α 

Α1.  

 1 – ΣΩΣΤΟ 

 2 – ΣΩΣΤΟ 

 3 – ΛΑΘΟΣ 

 4 – ΣΩΣΤΟ 

 5 – ΛΑΘΟΣ 

  

Α2.   

  ΔΙΑΒΑΣΕ Χ 

  ΑΝ TOP < 10 ΤΟΤΕ 

    TOP <-- TOP + 1 

    Α[TOP] <-- Χ 

  ΑΛΛΙΩΣ 

    ΓΡΑΨΕ 'ΥΠΕΡΧΕΊΛΙΣΗ ΣΤΟΙΒΑΣ' 

  ΤΕΛΟΣ_ΑΝ 

 

Α3.   

Σχολικό βιβλίο (Πληροφορική Συμπληρωματικό Εκπαιδευτικό Υλικό) σελίδα 42 

Α4.  

Σχολικό βιβλίο (Ανάπτυξη Εφαρμογών σε Προγραμματιστικό Περιβάλλον) σελίδα   

184 

ΘΕΜΑ Β 

Β1.  

 
 

Β2.   

Αλγόριθμος Β2   

  S <-- 0 

  Ι <-- 1 

  Όσο Ι <= 20 επανάλαβε 

    Αρχή_επανάληψης 

      Εμφάνισε 'Δώσε θετικό αριθμό' 

      Διάβασε Π[Ι] 

    Μέχρις_ότου Π[Ι] > 0 

    S <-- S + Π[Ι] 

    Ι <-- Ι + 1 

  Τέλος_επανάληψης 

  Εμφάνισε S 

Τέλος Β2 

 

Εναλλακτικά με χρήση ΓΙΑ: 

Αλγόριθμος Β2   

  S <-- 0 

  Για i Από 1 Μέχρι 20 

    Αρχή_επανάληψης 

      Εμφάνισε 'Δώσε θετικό αριθμό' 

      Διάβασε Π[Ι] 

    Μέχρις_ότου Π[Ι] > 0 

    S <-- S + Π[Ι] 

  Τέλος_επανάληψης 

  Εμφάνισε S 

Τέλος Β2 

  

 



 

3 
 

Β3.  

1 – ΛΟΓΙΚΗ 

2 – ΑΛΗΘΗΣ 

3 – j 

4 – i + j 

5 – 0 

6 – ΨΕΥΔΗΣ 

7 – f 

  

ΘΕΜΑ Γ 

ΠΡΟΓΡΑΜΜΑ ΘΕΜΑ_Γ 

ΣΤΑΘΕΡΕΣ 

  ΟΡΙΟ=10.30 

ΜΕΤΑΒΛΗΤΕΣ 

  ΑΚΕΡΑΙΕΣ:ΜΑΘΗΤΕΣ,ΠΡΟΚΡΙΘΗΚΑΝ,ΠΡΟΣΠΑΘΕΙΕΣ 

  ΠΡΑΓΜΑΤΙΚΕΣ:ΜΑΧ1,ΜΑΧ2,ΕΠ 

  ΧΑΡΑΚΤΗΡΕΣ:ΟΝ_ΜΑΧ1,ΟΝ_ΜΑΧ2,ΟΝ 

  ΛΟΓΙΚΕΣ:ΠΡΟΚΡΙΝΕΤΑΙ 

ΑΡΧΗ 

  ΜΑΘΗΤΕΣ <-- 0 

  ΠΡΟΚΡΙΘΗΚΑΝ <-- 0 

  ΜΑΧ1 <-- 0 

  ΜΑΧ2 <-- 0 

  ΟΝ_ΜΑΧ1 <-- ' ' 

  ΟΝ_ΜΑΧ2 <-- ' ' 

  ΓΡΑΨΕ 'ΔΩΣΤΕ ΟΝΟΜΑ' 

  ΔΙΑΒΑΣΕ ΟΝ 

  ΟΣΟ ΟΝ<>'ΤΕΛΟΣ' ΕΠΑΝΑΛΑΒΕ  !Γ2 

    ΜΑΘΗΤΕΣ <-- ΜΑΘΗΤΕΣ+1 

    ΠΡΟΣΠΑΘΕΙΕΣ <-- 0 

    ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ 

      ΠΡΟΚΡΙΝΕΤΑΙ <-- ΨΕΥΔΗΣ 

      ΓΡΑΨΕ 'ΔΩΣΤΕ ΕΠΙΔΟΣΗ' 

      ΔΙΑΒΑΣΕ ΕΠ 

      ΠΡΟΣΠΑΘΕΙΕΣ <-- ΠΡΟΣΠΑΘΕΙΕΣ+1 

      ΑΝ ΕΠ>ΟΡΙΟ ΤΟΤΕ 

        ΠΡΟΚΡΙΝΕΤΑΙ <-- ΑΛΗΘΗΣ 

      ΤΕΛΟΣ_ΑΝ 

    ΜΕΧΡΙΣ_ΟΤΟΥ ΠΡΟΚΡΙΝΕΤΑΙ=ΑΛΗΘΗΣ Ή ΠΡΟΣΠΑΘΕΙΕΣ=5 

    ΓΡΑΨΕ ΟΝ      !Γ3 

    ΑΝ ΠΡΟΚΡΙΝΕΤΑΙ=ΑΛΗΘΗΣ ΤΟΤΕ 

      ΓΡΑΨΕ 'ΠΡΟΚΡΙΝΕΤΑΙ',ΕΠ,ΠΡΟΣΠΑΘΕΙΕΣ 

      ΠΡΟΚΡΙΘΗΚΑΝ <-- ΠΡΟΚΡΙΘΗΚΑΝ+1 

      ΑΝ ΕΠ>ΜΑΧ1 ΤΟΤΕ    !Γ4 

        ΜΑΧ2 <-- ΜΑΧ1 

        ΟΝ_ΜΑΧ2 <-- ΟΝ_ΜΑΧ1 

        ΜΑΧ1 <-- ΕΠ 

        ΟΝ_ΜΑΧ1 <-- ΟΝ 

      ΑΛΛΙΩΣ_ΑΝ ΕΠ>ΜΑΧ2 ΤΟΤΕ 

        ΜΑΧ2 <-- ΕΠ 

        ΟΝ_ΜΑΧ2 <-- ΟΝ 

      ΤΕΛΟΣ_ΑΝ 

    ΑΛΛΙΩΣ 

      ΓΡΑΨΕ 'ΔΕΝ ΠΡΟΚΡΙΝΕΤΑΙ' 

    ΤΕΛΟΣ_ΑΝ 

    ΓΡΑΨΕ 'ΔΩΣΤΕ ΕΠΟΜΕΝΟ ΟΝΟΜΑ' 

    ΔΙΑΒΑΣΕ ΟΝ 

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΓΡΑΨΕ ΠΡΟΚΡΙΘΗΚΑΝ/ΜΑΘΗΤΕΣ*100 !Γ5 

  ΓΡΑΨΕ ΟΝ_ΜΑΧ1,ΜΑΧ1,ΟΝ_ΜΑΧ2,ΜΑΧ2 

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ 

 

Δεύτερη προτεινόμενη λύση: 

ΠΡΟΓΡΑΜΜΑ ΘΕΜΑΓ 

ΜΕΤΑΒΛΗΤΕΣ 

  ΑΚΕΡΑΙΕΣ:ΜΑΘΗΤΕΣ,ΠΡΟΚΡΙΘΗΚΑΝ,ΠΡΟΣΠΑΘΕΙΕΣ 

  ΠΡΑΓΜΑΤΙΚΕΣ:ΜΑΧ1,ΜΑΧ2,ΕΠ 

  ΧΑΡΑΚΤΗΡΕΣ:ΟΝ_ΜΑΧ1,ΟΝ_ΜΑΧ2,ΟΝ 

  ΛΟΓΙΚΕΣ:ΠΡΟΚΡΙΝΕΤΑΙ 

ΑΡΧΗ 

  ΜΑΘΗΤΕΣ <-- 0 

  ΠΡΟΚΡΙΘΗΚΑΝ <-- 0 

  ΜΑΧ1 <-- 0 

  ΜΑΧ2 <-- 0 

  ΟΝ_ΜΑΧ1 <-- ' ' 

  ΟΝ_ΜΑΧ2 <-- ' '        

  ΓΡΑΨΕ 'ΔΩΣΤΕ ΟΝΟΜΑ' 

  ΔΙΑΒΑΣΕ ΟΝ 

  ΟΣΟ ΟΝ<>'ΤΕΛΟΣ' ΕΠΑΝΑΛΑΒΕ  !Γ2 

    ΜΑΘΗΤΕΣ <-- ΜΑΘΗΤΕΣ+1 

    ΠΡΟΣΠΑΘΕΙΕΣ <-- 0 

    ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ 

      ΓΡΑΨΕ 'ΔΩΣΤΕ ΕΠΙΔΟΣΗ' 

      ΔΙΑΒΑΣΕ ΕΠ 

      ΠΡΟΣΠΑΘΕΙΕΣ <-- ΠΡΟΣΠΑΘΕΙΕΣ+1 

    ΜΕΧΡΙΣ_ΟΤΟΥ ΕΠ>10.3 Ή ΠΡΟΣΠΑΘΕΙΕΣ>5 

    ΓΡΑΨΕ ΟΝ      !Γ3 

    ΑΝ ΕΠ>10.3 ΤΟΤΕ 

      ΓΡΑΨΕ 'ΠΡΟΚΡΙΝΕΤΑΙ',ΕΠ,ΠΡΟΣΠΑΘΕΙΕΣ 

      ΠΡΟΚΡΙΘΗΚΑΝ <-- ΠΡΟΚΡΙΘΗΚΑΝ+1 

      ΑΝ ΕΠ>ΜΑΧ1 ΤΟΤΕ    !Γ4 

        ΜΑΧ2 <-- ΜΑΧ1 

        ΟΝ_ΜΑΧ2 <-- ΟΝ_ΜΑΧ1 

        ΜΑΧ1 <-- ΕΠ 

        ΟΝ_ΜΑΧ1 <-- ΟΝ 

      ΑΛΛΙΩΣ_ΑΝ ΕΠ>ΜΑΧ2 ΤΟΤΕ 

        ΜΑΧ2 <-- ΕΠ 

        ΟΝ_ΜΑΧ2 <-- ΟΝ 

      ΤΕΛΟΣ_ΑΝ 

    ΑΛΛΙΩΣ 

      ΓΡΑΨΕ 'ΔΕΝ ΠΡΟΚΡΙΝΕΤΑΙ' 

    ΤΕΛΟΣ_ΑΝ 

    ΓΡΑΨΕ 'ΔΩΣΤΕ ΕΠΟΜΕΝΟ ΟΝΟΜΑ' 

    ΔΙΑΒΑΣΕ ΟΝ 

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΓΡΑΨΕ ΠΡΟΚΡΙΘΗΚΑΝ/ΜΑΘΗΤΕΣ*100 !Γ5 

  ΓΡΑΨΕ ΟΝ_ΜΑΧ1,ΜΑΧ1,ΟΝ_ΜΑΧ2,ΜΑΧ2 

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ 

 

ΘΕΜΑ Δ 

ΠΡΟΓΡΑΜΜΑ ΘΕΜΑ_Δ   

ΜΕΤΑΒΛΗΤΕΣ            

  ΑΚΕΡΑΙΕΣ: i,j,Β[100],TEMP, Κ  

  ΧΑΡΑΚΤΗΡΕΣ:ΣΑ[30],ΟΝ[100],ΑΠ[100,30],TEMP_ΟΝ 

ΑΡΧΗ 

  ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 30 

    ΔΙΑΒΑΣΕ ΣΑ[i] 

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 100 

    ΔΙΑΒΑΣΕ ΟΝ[i] 

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 100 

    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 30 

      ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ 

        ΔΙΑΒΑΣΕ ΑΠ[i,j] 

      ΜΕΧΡΙΣ_ΟΤΟΥ ΑΠ[i,j] = 'Α' Ή ΑΠ[i,j] = 'Β' Ή ΑΠ[i,j] = 'Γ' 

    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 100 

    Β[i] <-- ΒΑΘΜΟΣ(ΑΠ,ΣΑ,i) 

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΓΙΑ i ΑΠΟ 2 ΜΕΧΡΙ 100 

    ΓΙΑ j ΑΠΟ 100 ΜΕΧΡΙ i ΜΕ ΒΗΜΑ-1 

      ΑΝ Β[j-1] < Β[j] ΤΟΤΕ 

        TEMP <-- Β[j-1] 

        Β[j-1] <-- Β[j] 

        Β[j] <-- TEMP 

        TEMP_ΟΝ <-- ΟΝ[j-1] 

        ΟΝ[j-1] <-- ΟΝ[j] 

        ΟΝ[j] <-- TEMP_ΟΝ 

      ΤΕΛΟΣ_ΑΝ 

    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10 

    ΓΡΑΨΕ ΟΝ[i] 

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  Κ <-- Β[10] 

  i <-- 11 

  ΟΣΟ Β[i] = Κ ΚΑΙ i <= 100 ΕΠΑΝΑΛΑΒΕ 

    ΓΡΑΨΕ ΟΝ[i] 

    i <-- i + 1 

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ  

 

ΣΥΝΑΡΤΗΣΗ ΒΑΘΜΟΣ(ΑΠ,ΣΑ,i):ΑΚΕΡΑΙΑ 

ΜΕΤΑΒΛΗΤΕΣ 

  ΑΚΕΡΑΙΕΣ:i,j,ΠΛ 

  ΧΑΡΑΚΤΗΡΕΣ:ΣΑ[30],ΑΠ[100,30] 

ΑΡΧΗ 

  ΠΛ <-- 0 

  ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 30 

    ΑΝ ΑΠ[i,j] = ΣΑ[j] ΤΟΤΕ 

      ΠΛ <-- ΠΛ + 1 

    ΤΕΛΟΣ_ΑΝ 

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΒΑΘΜΟΣ <-- ΠΛ*2 

ΤΕΛΟΣ_ΣΥΝΑΡΤΗΣΗΣ        

 

!ΕΝΑΛΛΑΚΤΙΚΑ ΓΙΑ ΤΟ Δ4 

ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10 

  ΓΡΑΨΕ ΟΝ[i] 

  ΑΝ i = 10 ΤΟΤΕ 

    j <-- 1                

    ΟΣΟ Β[i] = Β[i+j] ΕΠΑΝΑΛΑΒΕ 

      ΓΡΑΨΕ ΟΝ[i+j] 

      j <-- j + 1 

    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

  ΤΕΛΟΣ_ΑΝ 

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 


